

Csound

Csound

opcodes

v.1

pino@pinocappellano.it – www.pinocappellano.it

1

Csound – parametri

Parametri

Ogni campo delle Note può essere parametrato

p1 **p2** **p3** **p4** **p5** ...

i1 **0** **5.6**

Cosicché si può demandare allo score (e alle singole note) l'indicazione, per esempio, di frequenza e ampiezza.

In questo modo non c'è bisogno di creare tanti strumenti quante sono le note (con diversi parametri).

i1	indica il numero della nota (preceduto dal segno "i");
0	indica il momento (in secondi) in cui la nota viene creata;
5.6	indica la durata (in secondi) della nota

v.1

pino@pinocappellano.it – www.pinocappellano.it

2

Csound – parametri

Parametri

Esempio:

orchestra

```
; intestazione
sr = 44100
kr = 4410
ksmps = 10
nchnls = 1

; strumenti
instr 1
avariab oscil p4, p5, 1
out avariab
endin
```

score

```
; funzione
f1 0 8192 10 1

; note
i1 0 2 10000 440
i1 2 2 12000 220
i1 4 2 8000 110
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

3

ampdb ampdb(x)

Conversione

Converte in ampiezza lineare l'equivalente valore indicato in decibel.

- 60 dB = 1000
- 66 dB = 1995.262
- 72 dB = 3891.07
- 78 dB = 7943.279
- 84 dB = 15848.926
- 90 dB = 31622.764

```
instr 1
iamp = ampdb(p4)
aout oscil iamp, 440, 1
out aout
endin
```

```
i1 0 4 80
```

Esegue un LA a 80 dB

v.1

pino@pinocappellano.it – www.pinocappellano.it

4

cpspch cpspch(pch)

Conversione

Converte da **ottava.nota** in **hertz** (cicli per secondo = cps).

Do	Do#	Re	Re#	Mi	Fa	Fa#	Sol	Sol#	La	La#	Si
0	1	2	3	4	5	6	7	8	9	10	11

L'ottava è divisa in 12 parti (semitoni).
Ogni nota equivale al **dodicesimo** corrispondente (partendo da 0).

```
instr 1
ifreq = cpspch(p4)
aout oscil 10000, ifreq, 1
out aout
endin
```

Esegue il LA della settimana
ottava

```
i1 0 4 7.09
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

5

cpsoct cpsoct(oct)

Conversione

Converte da **ottava.decimale** in **hertz** (cicli per secondo = cps).

Do	Do#	Re	Re#	Mi	Fa	Fa#	Sol	Sol#	La	La#	Si
0	8,33	16,66	25	33,33	41,66	50	58,33	66,66	75	83,33	91,66

L'ottava è divisa in 100 parti.
Ogni nota equivale alla **frazione** corrispondente.

```
instr 1
ifreq = cpsoct(p4)
aout oscil 10000, ifreq, 1
out aout
endin
```

Esegue il LA della settimana
ottava

```
i1 0 4 7.75
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

6

oscil

var oscil xamp, xcps, ifn [,iphs]

Oscillatore semplice

Oscil simula un oscillatore e genera segnali audio (o segnali di controllo periodici) di determinata ampiezza (xamp o kamps) e frequenza (xcps), creati sulla base di una funzione (ifn) e con una certa fase (iphs).

```
instr 1
;      amp      freq      ident [opcion.
;      funz.    fase]
;      xamp,    xcps,    ifn    [,iphs]
aout  oscil 10000, 440, 1
out    aout
endin
```

```
f1 0 8192 10 1
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

7

oscili

var oscili xamp, xcps, ifn [,iphs]

Oscillatore semplice

oscili realizza un'interpolazione tra i valori della tabella. Di conseguenza si migliora la qualità del suono. Utilissimo quando si usano tabelle piccole (16, 32).

```
instr 1
;      amp      freq      ident [opcion.
;      funz.    fase]
;      xamp,    xcps,    ifn    [,iphs]
aout  oscil 10000, 440, 1
out    aout
endin
```

```
f1 0 8192 10 1
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

8

line


var line ia, idur1, ib

Glissando

Traccia una linea tra punti specifici

```
instr 1
;          freq   durata  freq
;          part.  arrivo
;nr       (line) ia,    idur1, ib
kfreq    line    p5,    p3,    p6
aout     oscil   p4,    kfreq, 1
         out    aout
         endin
```

```
f1 0 8192 10 1
i1 0 5 10000 440 3520
```



v.1

pino@pinocappellano.it – www.pinocappellano.it

9

expon


var expon ia, idur1, ib

Glissando esponenziale

Traccia una curva esponenziale tra punti specifici

```
instr 1
;          freq   durata  freq
;          part  arrivo
;nr       (expon) ia,    idur1, ib
kfreq    expon   p5,    p3,    p6
aout     oscil   p4,    kfreq, 1
         out    aout
         endin
```

```
f1 0 8192 10 1
i1 0 5 10000 440 3520
```



v.1

pino@pinocappellano.it – www.pinocappellano.it

10

linseg

var linseg ia, idur1, ib [,idur2, ic[...]]


Inviluppo

```
instr 1
;          amp  durata  amp  durata  amp  durata  amp  durata  amp
;          inizio fine      fine      fine      fine      fine
;nr  (linseg) ia,  idur1,  ib    [,idur2,ic    [,idur3,id    [,idur4,ie]]]
kenvlp linseg 0,    0.1,    p4*1.2, .2,    p4,    p3-1,  p4,    .7,    0
aout  oscil  kenvlp, p5,    1
out  aout
endin
```

```
f1 0 8192 10 1
i1 0 5 10000 440
```

Traccia una serie di linee tra punti specifici

La somma delle durate deve essere uguale a p3



v.1

pino@pinocappellano.it – www.pinocappellano.it

11

expseg

var expseg ia, idur1, ib [,idur2, ic[...]]


Inviluppo esponenziale

```
instr 1
;          amp  durata  amp  durata  amp  durata  amp  durata  amp
;          part  ar/par  ar/par  ar/par  ar/par  ar/par  ar/par  ar/par  arr
;          ia    idur1  ib    idur2  ic    idur3  id    idur4  ie
kenvlp expseg 0.001, 0.1,  p4*1.2, .2,    p4,    p3-1,  p4,    .7,    0.001
aout  oscil  kenvlp, p5,    1
out  aout
endin
```

```
f1 0 8192 10 1
i1 0 5 10000 440
```

Traccia una serie di segmenti esponenziali tra punti specifici

La somma delle durate deve essere uguale a p3



v.1

pino@pinocappellano.it – www.pinocappellano.it

12

Tremolo

Tremolo

L'ampiezza è modulata da un segnale di controllo che consiste in un ulteriore oscillatore.

La frequenza dell'oscillatore di controllo non rientra nell'ordine delle frequenze audio.

```
instr 1
kmodamp oscili p4, p5, 1
kamp = p4 + kmodamp
aout oscil kamp, 440, 1
out aout
endin
```

```
f1 0 8192 10 1
i1 0 5 10000 2
```

Vibrato

Vibrato

La frequenza è modulata da un segnale di controllo che consiste in un ulteriore oscillatore.

La frequenza dell'oscillatore di controllo non rientra nell'ordine delle frequenze audio.

```
instr 1
kmodfrq oscili p6, p7, 1
afreq = p5 + kmodfrq
aout oscil p4, afreq, 1
out aout
endin
```

```
f1 0 8192 10 1
i1 0 5 10000 440 140 3
```

soundin

var soundin ifilcod [,iskptim] [,iformat]

File esterno

Un file sonoro esterno può essere inserito con l'opcode **soundin** (non necessita di funzione). Parametri:
ifilcod = nome del file (se stringa: tra virgolette; se numero: il file sarà p.e. soundin.1)
Parametri opzionali:
iskptim = tempo, in secondi, prima dell'esecuzione del campione audio (default = 0)
iformat = indica il formato in bit (default = legge quello del file)

```
instr 1
a1 soundin "suono.wav"
out a1
endin
```

oppure:

```
instr 1
a1 soundin 1
out a1
endin
```

```
i1 0 3.3
```

```
i1 0 3.3
```

il file sorgente è "suono.wav"

il file sorgente è "soundin.1"

v.1

pino@pinocappellano.it – www.pinocappellano.it

15

diskin

var diskin ifilcod, kratio [,iskiptim] [,iwraparound] [,iformat]

File esterno

Un file sonoro esterno può essere inserito con l'opcode **diskin** (non necessita di funzione). Parametri:
ifilcod = nome del file (se stringa: tra virgolette; se numero: il file sarà p.e. diskin.1)
kratio = indica la variazione della frequenza di riproduzione (unità di misura: l'ottava)
Parametri opzionali:
iskiptim = tempo, in secondi, prima dell'esecuzione del campione audio (default = 0)
iwraparound = ripete (1) o non ripete (0) in loop la riproduzione (default = 0)
iformat = indica il formato in bit (default = legge quello del file)

```
instr 1
a2 diskin 1, p4
out a2
endin
```

```
i1 0 3.3 1 ;frequenza originale
i1 5 3.3 1.5 ;sopra la frequenza originale
i1 10 3.3 .5 ;metà della frequenza originale, un'ottava sotto
i1 15 3.3 -1 ;all'indietro, alla frequenza originale
i1 20 3.3 -2 ;all'indietro, un'ottava sopra
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

16

rand

var rand xamp [, iseed] [, isel] [, ioffset]

Numeri casuali (random)

rand genera una serie controllata di numeri casuali.

Parametri:

xamp = valore entro il quale vengono generati i numeri (da -xamp a +xamp).

iseed [opzionale, default=0.5] Valore iniziale (da 0 a 1) che viene moltiplicato al risultato.

isel [opzionale, default=0] Se 0, viene generato un numero a 16-bit; se diverso da 0, a 31 bit.

ioffset [opzionale, default=0] Valore che viene aggiunto al numero random generato.

Nell'esempio viene generato **Rumore Bianco**.

Per esso non c'è bisogno di alcuna funzione di tabella perché l'opcode genera forme d'onda casuali (non necessariamente, per esempio, sinusoidali).

```
instr 1
  a1 rand 20000
  out a1
endin
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

17

rand

var rand xamp [, iseed] [, isel] [, ioffset]

Numeri casuali (random)

rand genera una serie controllata di numeri casuali.

Parametri:

xamp = valore entro il quale vengono generati i numeri (da -xamp a +xamp).

iseed [opzionale, default=0.5] Valore iniziale (da 0 a 1) che viene moltiplicato al risultato.

isel [opzionale, default=0] Se 0, viene generato un numero a 16-bit; se diverso da 0, a 31 bit.

ioffset [opzionale, default=0] Valore che viene aggiunto al numero random generato.

Nell'esempio viene generato **Rumore a banda ristretta**.

```
instr 1
  ; Genera una frequenza casuale fra 140 e 740.
  kfreq rand 300
  kcps = kfreq + 440

  a1 oscil 20000, kcps, 1
  out a1
endin
```

```
f1 0 4096 10 1
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

18

randh

var randh xamp, xcps [, iseed] [, isize] [, ioffset]

Numeri casuali (random)

randh genera una serie di numeri casuali e li “tiene” per una certa durata.

Parametri:

xamp = valore entro il quale vengono generati i numeri (da -xamp a +xamp).

xcps = frequenza di generazione dei numeri

iseed [opzionale, default=0.5] Valore iniziale (da 0 a 1) che viene moltiplicato al risultato.

isel [opzionale, default=0] Se 0, viene generato un numero a 16-bit; se diverso da 0, a 31 bit.

ioffset [opzionale, default=0] Valore che viene aggiunto al numero random generato.

```
instr 1
; Genera una frequenza casuale fra 140 e 740 4 volte al secondo.
kfreq randh 300, 4
kcps = kfreq + 440

a1 oscil 20000, kcps, 1
out a1
endin
```

```
f1 0 4096 10 1
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

19

randi

var randi xamp, xcps [, iseed] [, isize] [, ioffset]

Numeri casuali (random)

randi genera una serie controllata di numeri random creando un'interpolazione con i nuovi numeri

Parametri:

xamp = valore entro il quale vengono generati i numeri (da -xamp a +xamp).

xcps = frequenza di generazione dei numeri

iseed [opzionale, default=0.5] Valore iniziale (da 0 a 1) che viene moltiplicato al risultato.

isel [opzionale, default=0] Se 0, viene generato un numero a 16-bit; se diverso da 0, a 31 bit.

ioffset [opzionale, default=0] Valore che viene aggiunto al numero random generato.

```
instr 1
; Genera una frequenza casuale fra 140 e 740 4 volte al secondo
; interpolando i risultati.
kfreq randi 300, 4
kcps = kfreq + 440

a1 oscil 20000, kcps, 1
out a1
endin
```

```
f1 0 4096 10 1
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

20

Onda a dente di sega

Sintesi additiva

Con la sintesi additiva si possono creare delle forme d'onda attraverso la somma di forme d'onda semplici, come le sinusoidi.

Onda a dente di sega: sono presenti tutte le armoniche

```
instr 1
  a1 oscil 10000, 440, 1
  out a1
endin
```

```
f1 0 4096 10 10 5 3.3 2.5 2 1.6 1.4 1.25 1.1 1
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

21

Onda quadra

Sintesi additiva

Con la sintesi additiva si possono creare delle forme d'onda attraverso la somma di forme d'onda semplici, come le sinusoidi.

Onda quadra: sono presenti solo le armoniche dispari

```
instr 1
  a1 oscil 10000, 440, 1
  out a1
endin
```

```
f1 0 4096 10 10 0 3.3 0 2 0 1.4 0 1.1
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

22

tone

var tone asig, khp [, iskip]

Sintesi sottrattiva

Con la sintesi sottrattiva si crea un nuovo suono sottraendo da quello originale alcune componenti. In genere vengono applicati al suono originale dei filtri.

tone è un opcode che funziona come **filtro passa-basso**. Parametri:

asig = segnale da filtrare

khp = frequenza di taglio (le frequenze al di sopra della soglia vengono tagliate)

iskip [opzionale, default=0] = Se 1 il filtro "lavora" anche sul campione precedente.

Nell'esempio viene filtrato del rumore bianco con un filtro passa-basso a 1000 Hz.

```
instr 1
a1 rand 20000
a2 tone a1, 1000
out a2
endin
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

23

atone

var atone asig, khp [, iskip]

Sintesi sottrattiva

Con la sintesi sottrattiva si crea un nuovo suono sottraendo da quello originale alcune componenti. In genere vengono applicati al suono originale dei filtri.

atone è un opcode che funziona come **filtro passa-alto**. Parametri:

asig = segnale da filtrare

khp = frequenza di taglio (le frequenze al di sopra della soglia vengono tagliate)

iskip [opzionale, default=0] = Se 1 il filtro "lavora" anche sul campione precedente.

Nell'esempio viene filtrato del rumore bianco con un filtro passa-alto a 19000 Hz.

```
instr 1
a1 rand 20000
a2 atone a1, 19000
out a2
endin
```

```
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

24

Modulazione di ampiezza

Modulazione di ampiezza ad anello

L'ampiezza è modulata da un segnale che consiste in un ulteriore oscillatore.

Nell'esempio, il segnale risultante ha 2 componenti frequenziali: 240 e 640 Hz.

```
instr 1
abimod oscili 5000, 200, 1
aport  oscil  abimod, 440, 1
out    aport
endin
```

```
f1 0 8192 10 1
i1 0 3
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

25

Modulazione di ampiezza

Modulazione di ampiezza (classica)

L'ampiezza è modulata da un segnale che consiste in un ulteriore oscillatore al quale viene sommato il valore dell'ampiezza.

Nell'esempio, il segnale risultante ha tre componenti frequenziali: 240, 440 e 640 Hz.

```
instr 1
abimod oscili 5000, 200, 1
aunimod = abimod+5000
aport  oscil  aunimod, 440, 1
out    aport
endin
```

```
f1 0 8192 10 1
i1 0 3
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

26

foscil

var foscil xamp, kcps, xcar, xmod, kndx, ifn [,iphs]

Modulazione di frequenza

foscil è un opcode che genera una modulazione di frequenza. Parametri:

xamp = ampiezza del segnale

kcps = frequenza di riferimento

xcar = fattore che, moltiplicato per kcps, restituisce la frequenza portante

xmod = fattore che, moltiplicato per kcps, restituisce la frequenza modulante

kndx = indice di modulazione (più e alto più aumenta il numero delle frequenze udibili)

ifn = numero della funzione della tabella utilizzata

iphs (opzionale, default=0) = fase iniziale (da 0 a 1)

Con la modulazione di frequenza vengono prodotte una serie teoricamente infinita di frequenze.

```
instr 1
;      kcamp kcps kcar xmod kndx ifn
aout foscil p4, p5, p6, p7, p8, 1
out aout
endin
```

```
f1 0 8192 10 1
;
; instr act dur      kcamp   kcps      xcar      xmod      kndx
;      amp.   frq.base  fatt.frq.port.  fatt.frq.mod.  indx.modul.
i1  0   3   3   10000   1         100         50         10
i1  3   3   3   10000   1         100         50         20
i1  6   3   3   10000   1         100         50         30
i1  9   3   3   10000   1         100         50         50
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

27

grain

var grain xamp, xcps, xdens, kampoff, kcpsoff, kgdur, igfn, iwfn, imgdur [,igrnd]

Sintesi granulare

Parametri:

xamp = ampiezza di ogni grano

xcps = frequenza dei grani:

xdens = densità: numero di grani per secondo

kampoff = massima deviazione (random) dell'ampiezza dei grani (se 0 = nessuna variazione)

kcpsoff = massima deviazione (random) della frequenza dei grani

kgdur = durata dei grani in secondi

igfn = Numero della funzione della tabella che crea i grani

iwfn = Numero della funzione della tabella che genera l'involuppo dell'ampiezza dei grani

imgdur = Massima durata dei grani in secondi

igrnd [opzionale] = di default è 0 e indica che i grani vengono generati dall'inizio di **iwfn**

La tabella creata dalla funzione f2 utilizza il GEN 19 oppure il GEN 20. Il GEN 19, dell'esempio, genera onde composte dalla somma di semplici sinusoidi "pesate". I parametri del GEN 19 sono:
pna [pnb...] = parz. di sinusoide
stra [strb...] = forza di pna...
phsa [phsb...] = fase in gradi di pna...
dcoa [dcob...] = DC offset di pna...

```
instr 1
;      xamp xcps xdens kampoff kcpsoff kgdur igfn iwfn imgdur
a1 grain 10000, 500, 30, 10000, 200, 0.05, 1, 2, 0.5
out a1
endin
```

```
f1 0 4096 10 1
;      pna stra phsa dcoa pnb strb phsb dcob ...
f2 0 4096 19 1 .5 270 .5
i1 0 5
```

v.1

pino@pinocappellano.it – www.pinocappellano.it

28